

# MAKING A MULTI CART TUTORIAL

As described on page 10 of *Using XB GDP.pdf*, it is possible to combine multiple cartridges into a single cartridge. Probably the most common use for this is to make a package containing several programs that can be selected from a main menu. This tutorial shows how to do that. The finished cartridge will display a bit mapped picture and let you select from 3 compiled games, Aperture, Night Stalker, and Robots of Death.

You should use XB 2.9 G.E.M. for this. As usual, we will have the XB Game Developer's Package in disk 1. For the working disk used in making the cartridges, we will use the folder MULTITUTOR as DSK4. Initially this folder contains:

APERTURE-X	compiled version of Aperture	by Adamantyr	(loads in XB)
STALKER-X	compiled version of Night Stalker	by LASooner	(loads in XB)
ROBOT-X	compiled version of Robots of Death	by Retrospect	(loads in XB)
BMMENU	menu program that loads bit mapped picture		(XB, uses TML)
POLYMENU	alternate menu program that uses polyspirals		(XB, uses TML)
HUMBIRD_P and HUMBIRD_C	bit mapped picture of a hummingbird		(TI Artist format)

The three games have already been compiled, and can be run from XB with RUN "DSK4.APERTURE-X", RUN "DSK4.STALKER-X", and RUN "DSK4.ROBOT-X".

To start with, we need a menu program that lets the user select from the three choices. The program below would be a very simple way to do this:

```
10 CALL CLEAR :: PRINT "1 FOR APERTURE": "2 FOR NIGHTSTALKER": "3 FOR ROBOTS OF DEATH"
20 CALL KEY(0,K,S):: IF K<49 OR K>51 THEN 20
30 ON K GOTO 40,50,60
40 RUN "DSK4.APERTURE-X"
50 RUN "DSK4.STALKER-X"
60 RUN "DSK4.ROBOT-X"
```

But there is no reason for the menu to be that simple. The menu program is just that – a program, and of course, programs can actually do things besides display text. So you are not restricted to a simple menu that just offers a list of the programs to load. Let's create something a little more interesting. Running under *The Missing Link*, we can load a bit mapped picture, print a menu in the lower left corner, use the space bar to highlight the desired program, and press enter to run the selected program. The picture is of a hummingbird, which has nothing to do with the cartridge other than to show how this can be done. You would choose a picture that actually has something to do with the theme of the cartridge. As usual, we want to be able to test this from Extended BASIC before compiling.

First load *The Missing Link*. This is option 6 on the main menu for XB 2.9 G.E.M., or you can CALL TML. Then OLD DSK4.BMMENU which will load the program below.

```
90 !CALL LINK("TML16")
100 CALL SCREEN(2)!set screen to black
110 CALL LINK("LOADP","DSK4.HUMBIRD")!load picture of hummingbird
120 CALL LINK("WINDOW",166,1,192,92):: CALL LINK("PENHUE",2,15):: CALL LINK("PE"):: CALL LINK("FILL")!
set window boundaries, colors, and clear window
130 CALL LINK("PRINT",2,1,"APERTURE")
140 CALL LINK("PRINT",10,1,"NIGHT STALKER")
150 CALL LINK("PRINT",18,1,"ROBOTS OF DEATH")
160 CALL LINK("PU"):: R=1 !from now on we will only be changing colors
```

```

170 CALL LINK("PENHUE",2,16):: CALL LINK("FILL",R,1,R+8,92)!hilight menu selection to black on white
180 CALL KEY(0,K,S):: IF K<>32 AND K<>13 OR S<1 THEN 180
190 CALL LINK("PENHUE",2,15):: CALL LINK("FILL",R,1,R+8,92)!unselect line with black on grey
200 IF K=13 THEN 230
201 !if enter then run the program
210 R=R+8 :: IF R>18 THEN R=1
211 !update the row
220 GOTO 170
230 CALL LOAD(-31804,0,0):: CALL LINK("GRAFIX"):: R=INT(R/8)+1 :: ON R GOTO 240,250,260 !turn off
interrupt, select graphics mode, make R from 1 to 3, and run
240 RUN "DSK4.APERTURE-X"
250 RUN "DSK4.STALKER-X"
260 RUN "DSK4.ROBOT-X"
Then SAVE DSK4.BMMENU

```

Test it out to be sure it looks as expected and that you can load and run the desired game.

Now we have an XB loader program that can load 3 compiled games. It is time to turn this into a cartridge.

The first step is to modify the loader program for compatibility with the compiler.

You want to do this in standard XB. First quit, then press 2 for XB. At the XB GDP menu, select XB, then enter:

```
OLD DSK4.BMMENU
```

When the program is loaded, add this line, which will initialize TML at startup:

```
90 CALL LINK("TML16")
```

Disk access is not used in a cartridge, so CALL LINK("LOADP") cannot be used. A different method will be used to copy it from ram into the VDP memory. (The picture will be embedded in the cartridge in a later step.) Replace line 110 with:

```
110 CALL MOVE(3,-24530,8192,6144):: CALL MOVE(3,-18386,0,6144)
```

The compiled code cannot run a program from disk. Instead, the last character of the filename tells it which banks of the cartridge to use when loading. Change lines 240, 250, and 260 to:

```

240 RUN "DSK4.B"           or           240 RUN "B"
250 RUN "DSK4.C"           or           240 RUN "C"
260 RUN "DSK4.D"           or           240 RUN "D"

```

The file names can be anything as long as they end with B, C, and D.

Now convert the CALL LINKs to lower case. This tells the compiler they are A/L subroutines.

```
CALL LOAD("DSK1.UC2LC")
```

```
CALL LINK("X")           (you can list the program to verify that the CALL LINKs have been converted)
```

```
SAVE DSK4.BMMENUL       (Append the L to show that it has been converted to lower case.)
```

At this point the compilation process proceeds normally. Keep pressing enter until you come to the loader. *N.B. You cannot do this from the 40 or 80 column editor.*

The last step if the compilation process is the loader. When it starts you are asked:

```

Using Assembly Support?    press Y
Compiled file to load?     DSK4.BMMENUL.OBJ
Assembly routines to load? DSK1.TMLC

```

Press Enter when prompted.

Now the compiled program and *The Missing Link* routines are loaded. Since we are making a cartridge, there is no need to save as an EA5 or XB program. Skip those steps with Fctn 3, then Enter.

At this point the program is loaded and you are in the XB line editor:

CALL FILES(1)                   to give us more ram space  
CALL LOAD(-31868,0,0,0,0)   makes XB run the program from VDP ram  
RUN "DSK1.MAKECART8"

For:

Using TML-add picture?       press Y  
Name of picture file"       DSK4.HUMBIRD  
Program name on menu?       COLIBRI  
Initialize VDP?           press Y  
Name of file?           DSK4.COLIBRIA-8.BIN       the A makes this the first cartridge in the chain

When finished, the cartridge containing the menu program has been created.

Now it is time to convert the three games into cartridges. These will be much easier because they are already compiled, and the only thing left is to load and convert them into cartridges.

From Extended BASIC do these steps for each of the three games:

NEW

CALL FILES(1)  
OLD DSK4.APERTURE-X                   this will be cartridge B  
      2<sup>nd</sup> pass       OLD DSK4.STALKER-X       this will be cartridge C  
      3<sup>rd</sup> pass       OLD DSK4.ROBOT-X       this will be cartridge D

Now the program has been loaded into memory.

CALL LOAD(-31868,0,0,0,0)   makes XB run the program from VDP ram  
RUN "DSK1.MAKECART8"

The program asks:

Program name on menu?   APERTURE                   (cartridge B)  
      2<sup>nd</sup> pass       NIGHT STALKER           (cartridge C)  
      3<sup>rd</sup> pass       ROBOTS OF DEATH       (cartridge D)  
Initialize VDP?       Y       (These are stand alone programs, so the VDP should be initialized)  
Name of file?       DSK4.COLIBRIB-8.BIN       (B makes this the second cartridge in the chain)  
      2<sup>nd</sup> pass       DSK4.COLIBRIC-8.BIN       (C makes this the third cartridge in the chain)  
      3<sup>rd</sup> pass       DSK4.COLIBRID-8.BIN       (D makes this the fourth cartridge in the chain)

The three game cartridges have been made. You can try them out to be sure they work as expected.

Now all four cartridges have been created, named: COLIBRIA-8 to COLIBRID-8. We can now make the multi cartridge.

NEW


RUN "DSK1.CHAINCARTS"

For:


Name of first cart in chain?   DSK4.COLIBRIA-8.BIN  
Name to put on the menu?   COLIBRI  
Name of chained cartridge?   DSK4.COLIBRI-8.BIN

The cartridges are combined into a single cartridge named COLIBRI-8.BIN, starting with COLIBRIA, then COLIBRIB, COLIBRIC, COLIBRID, and ending when COLIBRIE cannot be found.

The folder includes an alternate menu program called POLYMENU. Like the previous menu program, this puts the three options on the screen. You use the space bar to choose and press Enter to run the selected program. But instead of a bit mapped picture, this menu displays polyspirals on the screen until you make your selection. As a practice exercise, I will leave it to you to convert this into a cartridge by following the steps above. Since you will not be loading a picture, you will press N when given that option.



BMMENU



POLYMENU